

第12回 バイナリデータの解析

目標

- ・ バイナリデータの入出力と解析を学ぶ

0. 準備

今日の作業をするディレクトリを作成し、移動いなさい。

```
% mkdir 20180111
```

```
% cd 20180111
```

1. バイナリデータの入出力

これまではテキストデータを扱ってきた。ここではバイナリデータの取扱を学ぶ。

バイナリデータは2進数（0 と 1）の集まりで構成されている（テキストデータも 0 と 1 の集合体ではあるが、文字コードに対応した文字としての意味を持っており、テキストエディタで開けるという点で異なる）。バイナリデータを取り扱う際には、データがどのように格納されているかを知っておく必要がある。画像データの場合に必要な情報は画像の縦横サイズ、バイト数、ヘッダーサイズなどである。

今回は以下の月探査データを使用する。

- 1) 月周回衛星「かぐや」地形カメラ全球マップデータ [Haruyama et al. EPS 2008]

■ファイル名：tco_720x360.raw

画像サイズ：720 (横ピクセル数) x 360 (縦ピクセル数)

1 ピクセルのバイト数：1 (8 bit)

ヘッダーサイズ：0

http://www.eps.nagoya-u.ac.jp/~morota/tco_720x360.raw

- 2) 月周回衛星「かぐや」全球地殻厚マップデータ [Ishihara et al. GRL 2009]

■ファイル名：crust-basalt_mac_720x360.raw

画像サイズ：720 (横ピクセル数) x 360 (縦ピクセル数)


1 ピクセルのバイト数：2 (16 bit)

ヘッダーサイズ：0

バイト順序：intel


http://www.eps.nagoya-u.ac.jp/~morota/crust-basalt_mac_720x360.raw

これらは、緯度方向は-89.75度を始点として0.5度おき（ $180\text{度} \div 360\text{ピクセル} = 0.5\text{度/ピクセル}$ ）に89.75度まで、経度方向には0.25度から0.5度おき（ $360\text{度} \div 720\text{ピクセル} = 0.5\text{度/ピクセル}$ ）に359.75度までのデータとなっている。



■ 全球マップデータ (tco_720x360.raw)

様々な時期、撮像条件でえられた画像データをモザイクして作成。真中半分が月の裏側、両端が月の表側である。月の表側に海と呼ばれるマグマが噴出してできた領域が集中していることが分かる。



■ 全球地殻厚マップデータ (crust-basalt_mac_720x360.raw)

「かぐや」衛星の軌道のふらつきから月の重力異常の分布を調べ、重力異常を説明するように地殻とマントル境界（モホ面）の起伏を算出したもの。黒いほど地殻が薄いことを表している。真中下側にある大きな黒い領域は、太陽系最大の衝突クレータ、南極-エイトケン盆地である。

ImageMagick のコマンドを使用して、RAW ファイルから JPEG ファイルへの変換が可能である。JPEG ファイルに変換することによりプレビューなどの簡易画像表示ソフトで中身をチェックすることができる。

```
%convert -depth (bit 数) -size 720x360 -auto-level gray:(RAW ファイル名)
(JPEG ファイル名)
```

また、以下のコマンドで画像の上下反転ができる。-depth の後はバイト数ではなく bit 数であることに注意。「gray」のあとの「:」とファイル名の間にはスペースを入れないこと。

```
%convert -flip (元の JPEG ファイル名) (反転後の JPEG ファイル名)
```

練習問題 1

ファイル容量を確認し、1 ピクセルのデータバイト数、画像サイズ（縦と横のピクセル数）、ヘッダーサイズから計算されるファイルサイズと比較せよ。ファイルサイズは下記式で計算。
(ファイルサイズ) = (1 ピクセルのデータバイト数) × (縦ピクセル数) × (横ピクセル数) + (ヘッダーサイズ)

Fortran でのバイナリファイルの入出力にはテキストファイルと同様に read 文、write 文を使用する。以下に入出力を行ったプログラムの例を示す。

プログラムは

<http://www.eps.nagoya-u.ac.jp/~morota/20161222/program01.html>

からコピーできる

```

*****
!      program01
      implicit none
      integer size1,size2,hdr_size,line,sample
      integer poi1,poi2,p,q
      integer*2 CT(360,720)
      byte TCO(360,720)

      hdr_size=0          !*** ヘッダーサイズ
      sample=720          !*** 画像の横サイズ
      line=360            !*** 画像の縦サイズ
      size1=1             !*** TC データバイト数
      size2=2             !*** 地殻厚データバイト数

      open(10,file='./tco_720x360.raw', &
           form='unformatted',access='direct',recl=size1)
      open(20,file='./crust-basalt_mac_720x360.raw', &
           form='unformatted',access='direct',recl=size2)
      open(30,file='./tco_720x360-02.raw', &
           form='unformatted',access='direct',recl=size1)

      poi1=hdr_size+1     !*** データ部の頭を初期詠込位置に設定
      poi2=hdr_size+1     !*** データ部の頭を初期詠込位置に設定

      do p=1,line
        do q=1,sample
          read(10,rec=poi1) TCO(p,q)  !*** データの読み込み
          read(20,rec=poi2) CT(p,q)   !*** データの読み込み
          poi1=poi1+1
          poi2=poi2+1
        enddo
      enddo

      close(10)
      close(20)

      poi1=hdr_size+1     !*** データ部の頭を初期詠込位置に設定

      do p=1,line
        do q=1,sample
          write(30,rec=poi1) TCO(p,q)      !*** データの出力
          poi1=poi1+1
        enddo
      enddo

      close(30)

      end
*****

```

練習問題2

上記プログラムで出力されたファイル (tco_720x360-02.raw) を ImageMagick の convert コマンドを使って JPEG に変化し、元データと同じであることを確認せよ。

赤道の地殻厚のプロファイルをテキストデータとして出力するプログラムを以下に示す。

プログラムは

<http://www.eps.nagoya-u.ac.jp/~morota/20161222/program02.html>

からコピーできる

```
*****
!      program02
      implicit none
      integer size2,hdr_size,line,sample
      integer poi2,p,q
      integer*2 CT(360,720)
      real*8 long,dlong

      hdr_size=0                      !*** ヘッダーサイズ
      sample=720                     !*** 画像の横サイズ
      line=360                       !*** 画像の縦サイズ
      size2=2                        !*** 地殻厚データバイト数
      dlong=0.5D0                    !*** 経度のデータ間隔 [deg]

      open(20,file='./crust-basalt_mac_720x360.raw', &
           form='unformatted',access='direct',recl=size2)
      open(1,file='profile02.dat')

      poi2=hdr_size+1                !*** データ部の頭を初期詠込位置に設定

      do p=1,line
        do q=1,sample
          read(20,rec=poi2) CT(p,q)  !*** データの読み込み
          poi2=poi2+1
        enddo
      enddo

      close(20)

      long=0.25D0
      do q=1,sample
        write(1,*) long,dble((CT(180,q)+CT(181,q)))/2.D0
        long=long+dlong
      enddo

      close(1)

      end
*****
```

練習問題 3

上記のプログラムの出力結果を gnuplot でグラフにして, 月の表側 (経度 0~90 度, 270~360 度) と裏側 (経度 90~270 度) で地殻厚がどのように違うか考察せよ.

練習問題 4

上記のプログラムを書き換えて, 緯度方向の平均地殻厚プロファイルを出力するプログラムを作成せよ. また, 出力結果をグラフ化し, 月の地殻が赤道と極でどのように異なるか考察せよ.

2. 宿題

練習問題 4 の答えと考察, プログラムを下記アドレスに送ること.

宿題の提出先: 城野 (sirono@eps.nagoya-u.ac.jp)

諸田 (morota@eps.nagoya-u.ac.jp)

柴田 (shibata.takuma@h.mbox.nagoya-u.ac.jp)

宿題のしめきり: 1 月 16 日 (火曜日)

3. ログアウト

作業が終了したら必ずログアウトすること.

・ログアウト

「🍏 マーク」 => 「...をログアウト」